Mankiw, Romer, and Weil (1992 QJE)

[Knight-1946-AER]_ notes the need for a broad concept of capital. The Knightian notion is much broader than that used in the standard Solow model. [Mankiw.Romer.Weil-1992-QJE]_ begin to address this by showing that some empirical shortcomings of the Solow model are ameliorated by the addition of human capital. MRW's work received a lot of attention for several reasons:

- There had been a growing tendency to presume that the Solow model was unsatisfactory because it appeals to exogenous technology differences to explain cross country variations in labor productivity.
- However, and most provocatively, MRW argue that saving and population growth differences do a good job in explaining differences in per capita income.
- Economists were also interested in it as a counterthrust to the endogenous growth literature, which tends to assume constant returns to scale in capital inputs (possibly physical and human capital inputs).
- Finally, it offers an answer to a question raised by Romer: how to reconcile relatively similar growth rates with quite different per capita incomes.

MRW's key extension of the Solow growth model is the inclusion of the accumulation of human capital.

Predictions:

Recall the Solow model predicted that steady-state income per capita is higher when the saving rate (s) is higher but lower when population growth (g_N) is higher. MRW show that cross section data support these predictions, accounting for more than 50% of the variation in cross section data. Furthermore, MRW don't reject the Solow model prediction that effects of s and g_N are of equal magnitude but opposite in sign. However, the correlations in the data are larger than those predicted by the Solow model.

[Mankiw.Romer.Weil-1992-QJE]_ attempt to remedy this shortcoming by including a proxy for human capital (H) in a constant returns to scale production function, $F[K, H, L]$, where $L = A\,N$ is once again the effective labor supply. Let $h = H/L$ so that

$$\texttt{y==Y/L==(1/L)F[K,H,L]==F[k,h,1]==f[k,h]} \tag{1}$$

Model human capital accumulation just like physical capital accumulation, except that the resources devoted to each may differ. For notational simplicity, use a common depreciation rate.

$$\begin{aligned} \dot{K} &= s_k\, Y - d\, K \\ \dot{H} &= s_h\, Y - d\, H \end{aligned} \tag{2}$$

For notational simplicity, let $n = d + g_A + g_N$. Then

$$\dot{h} = s_h\, y - n\, h \tag{3}$$

The model is otherwise unchanged, so the dynamics of accumulation for physical capital remain the same.

$$\dot{k} = s_k\, y - n\, k \tag{4}$$

Characterization of the steady state becomes somewhat more involved, since there are two dynamic equations. If $\dot{k} == 0$ then $f[k, h]/k == n/s_k$. If $\dot{h} == 0$ then $f[k, h]/h == n/s_h$. In a steady state, both are true.

Exercise

Derive $f[k, h]$ from the linear homogeneous production function $F[K, H, L]$.

A Solution:

```
In[◦]:= crts = F[λ K, λ H, λ L] == λ F[K, H, L];
crts /. {λ → 1 / L}
% /. {K / L → k, H / L → h}
```

Out[◦]= $F\left[\frac{K}{L}, \frac{H}{L}, 1\right] == \frac{F[K, H, L]}{L}$

Out[◦]= $F[k, h, 1] == \frac{F[K, H, L]}{L}$

Steady State: Algebraic Details with Cobb-Douglas Production

Adopt a Cobb-Douglas production function: $F[K, H, L] = B\, K^\alpha\, H^\beta\, L^{1-\alpha-\beta}$, where $L = A\, N$. (Remember, our N is MRW's L.) For algebraic convenience, set $B = 1$ so that $f[k, h] = k^\alpha\, h^\beta$. To examine the implied steady state, set $\dot{h} == 0$ and $\dot{k} == 0$.

$$\begin{aligned} s_h\, k^\alpha\, h^\beta &\overset{ss}{==} n\, h \;\Rightarrow\; k^\alpha\, h^{\beta-1} == \frac{n}{s_h} \\ s_k\, k^\alpha\, h^\beta &\overset{ss}{==} n\, k \;\Rightarrow\; k^{\alpha-1}\, h^\beta == \frac{n}{s_k} \end{aligned} \tag{5}$$

Note for later that changing s_k only affects the $\dot{k} == 0$ equation, while changing s_h only affects the $\dot{h} == 0$ equation. However, a change in n affects both equations.

Taking logarithms produces a log-linear system, making it easy to solve for the steady-state values of Log[k] and Log[h].

$$\begin{aligned} \alpha\, \mathrm{Log}[k] - (1-\beta)\, \mathrm{Log}[h] &== \mathrm{Log}[n/s_h] \\ \beta\, \mathrm{Log}[h] - (1-\alpha)\, \mathrm{Log}[k] &== \mathrm{Log}[n/s_k] \end{aligned} \tag{6}$$

(Note that, since labor matters for production, $\alpha < 1 - \beta$ and $\beta < 1 - \alpha$.) Express this steady-state system as a matrix equation.

$$\begin{pmatrix} \alpha & -(1-\beta) \\ -(1-\alpha) & \beta \end{pmatrix} \begin{pmatrix} \mathrm{Log}[k] \\ \mathrm{Log}[h] \end{pmatrix} == \begin{pmatrix} \mathrm{Log}[n/s_h] \\ \mathrm{Log}[n/s_k] \end{pmatrix} \tag{7}$$

Finally, solve for Log[k] and Log[h] as

$$\begin{pmatrix} \text{Log}[k] \\ \text{Log}[h] \end{pmatrix} == \frac{-1}{1-\alpha-\beta} \begin{pmatrix} \beta & (1-\beta) \\ (1-\alpha) & \alpha \end{pmatrix} \begin{pmatrix} \text{Log}[n/s_h] \\ \text{Log}[n/s_k] \end{pmatrix} \tag{8}$$

Writing the system solution by hand and simplifying produces the following.

$$\begin{pmatrix} \text{Log}[k] \\ \text{Log}[h] \end{pmatrix} == \frac{1}{1-\alpha-\beta} \begin{pmatrix} \beta\,\text{Log}[s_h] + (1-\beta)\,\text{Log}[s_k] - \text{Log}[n] \\ (1-\alpha)\,\text{Log}[s_h] + \alpha\,\text{Log}[s_k] - \text{Log}[n] \end{pmatrix} \tag{9}$$

Note the weighted average of the logged saving rates in each case. Since $\text{Log}[y] = \alpha\,\text{Log}[k] + \beta\,\text{Log}[h]$, use these solutions to solve for steady-state income. This is our reduced-form expression for steady-state “per capita” income.

$$\text{Log}[y] = \frac{\alpha}{(1-\alpha-\beta)}\,\text{Log}[s_k] + \frac{\beta}{(1-\alpha-\beta)}\,\text{Log}[s_h] - \frac{(\alpha+\beta)}{(1-\alpha-\beta)}\,\text{Log}[n] \tag{10}$$

Exercise

Use Mathematica to solve the steady-state system.

Hint:

Use the **PowerExpand** and **Simplify** commands to simply your expressions.

A Solution:

There are many ways to solve this system. For example, we may use the **LinearSolve** command.

```
In[ ]:= mA = ( α        -(1 - β) ); (* coefficient matrix *)
             ( -(1 - α)   β      )
        vb = {Log[n / sh], Log[n / sk]}; (* exogenous vector *)
        (sln = LinearSolve[mA, vb]) // Apart // Column (* solve for Log[k] & Log[h] *)
```

Out[ ]=

$$\frac{\beta\,\text{Log}\left[\frac{n}{sh}\right]}{-1+\alpha+\beta} - \frac{(-1+\beta)\,\text{Log}\left[\frac{n}{sk}\right]}{-1+\alpha+\beta}$$

$$-\frac{(-1+\alpha)\,\text{Log}\left[\frac{n}{sh}\right]}{-1+\alpha+\beta} + \frac{\alpha\,\text{Log}\left[\frac{n}{sk}\right]}{-1+\alpha+\beta}$$

It is possible to simplify this expression.

```
In[ ]:= {α, β}.sln // PowerExpand // Simplify
```

Out[ ]=

$$\frac{(\alpha+\beta)\,\text{Log}[n] - \beta\,\text{Log}[sh] - \alpha\,\text{Log}[sk]}{-1+\alpha+\beta}$$

Alternatively, solve the same system using the inverse of the coefficient matrix.

```
In[•]:= (mAI = Inverse[mA]) // MatrixForm
```

$$mAI == \frac{-1}{1-\alpha-\beta}\begin{pmatrix}\beta & (1-\beta) \\ (1-\alpha) & \alpha\end{pmatrix} \text{ // Simplify}$$

Out[•]//MatrixForm=

$$\begin{pmatrix}\frac{\beta}{-1+\alpha+\beta} & \frac{1-\beta}{-1+\alpha+\beta} \\ \frac{1-\alpha}{-1+\alpha+\beta} & \frac{\alpha}{-1+\alpha+\beta}\end{pmatrix}$$

Out[•]= True

Finally, solve for Log[k] and Log[h] as

$$\begin{pmatrix}\text{Log}[k] \\ \text{Log}[h]\end{pmatrix} == \frac{-1}{1-\alpha-\beta}\begin{pmatrix}\beta & (1-\beta) \\ (1-\alpha) & \alpha\end{pmatrix}\begin{pmatrix}\text{Log}[n/s_h] \\ \text{Log}[n/s_k]\end{pmatrix} \tag{11}$$

In[•]:=

$$\begin{pmatrix}\text{Log}[k] \\ \text{Log}[h]\end{pmatrix} == \text{mAI}.\begin{pmatrix}\text{Log}[n / sh] \\ \text{Log}[n / sk]\end{pmatrix} \text{ // PowerExpand // Simplify;}$$

```
MatrixForm /@ %
```

Out[•]=

$$\begin{pmatrix}\text{Log}[k] \\ \text{Log}[h]\end{pmatrix} == \begin{pmatrix}\frac{\text{Log}[n]-\beta\,\text{Log}[sh]+(-1+\beta)\,\text{Log}[sk]}{-1+\alpha+\beta} \\ \frac{\text{Log}[n]+(-1+\alpha)\,\text{Log}[sh]-\alpha\,\text{Log}[sk]}{-1+\alpha+\beta}\end{pmatrix}$$

Visualizing Dynamic Adjustments and the Comparative Statics

We found that when $\dot{k} == 0$,

$$\beta\,\text{Log}[h] - (1-\alpha)\,\text{Log}[k] == \text{Log}[n/s_k]$$

so that

$$\text{Log}[k] == \text{Log}[h]\,\beta/(1-\alpha) - \text{Log}[n/s_k]/(1-\alpha)$$

So if we think of k as a function of h along the $\dot{k} == 0$ locus, we find that this locus has a constant elasticity of $\beta/(1-\alpha) < 1$. That is, an increase in h leads to a less than proportional increase in k (along this locus).

Symmetrically, along the $\dot{h} == 0$ locus,

$$\alpha\,\text{Log}[k] - (1-\beta)\,\text{Log}[h] == \text{Log}[n/s_h] \tag{12}$$

so that

$$\text{Log}[k] == \text{Log}[h]\,(1-\beta)/\alpha + \text{Log}[n/s_h]/\alpha \tag{13}$$

So if we think of k as a function of h along the $\dot{h} == 0$ locus, we find that this locus has a constant elasticity of $(1-\beta)/\alpha > 1$. That is, an increase in h leads to a more than proportional increase in k (along this locus).

Recall that $\alpha < 1-\beta$ and $\beta < 1-\alpha$. This ensures that along $\dot{k} == 0$, k is convex in h, and that similarly along $\dot{h} == 0$, h is convex in k.

Out[∘]=



Out[∘]=



If we increase s_k, then ceteris paribus a point that was on the $\dot{k}$ == 0 locus is now a point of positive capital accumulation. This gives us a shift up of the $\dot{k}$ == 0 locus, following by increases in k and h until the new steady state is reached.

If we increase s_h, then ceteris paribus a point that was on the $\dot{h}$ == 0 locus is now a point of positive human capital accumulation. This gives us a shift right of the $\dot{h}$ == 0 locus, following by increases in k and h until the new steady state is reached.

Exercise

Find the the elasticity of k with respect to h along the $\dot{k}$ == 0 locus. Find the the elasticity of k with respect to h along the $\dot{h}$ == 0 locus.

Hint:

Make this easier for Mathematica by first using a logarithmic transformation. (Use **Log** followed by the

PowerExpand command.)

A Solution:

In[∘]:=
```
kdot0 = Log[k^(α - 1) h^β] == Log[n / sk] // PowerExpand
Solve[kdot0, Log[k]]
D[Log[k] /. First@%, Log[h]]
```

Out[∘]= $\beta \,\text{Log}[h] + (-1 + \alpha)\,\text{Log}[k] == \text{Log}[n] - \text{Log}[sk]$

Out[∘]= $\left\{\left\{\text{Log}[k] \to \frac{-\beta\,\text{Log}[h] + \text{Log}[n] - \text{Log}[sk]}{-1 + \alpha}\right\}\right\}$

Out[∘]= $-\frac{\beta}{-1 + \alpha}$

In[∘]:=
```
hdot0 = Log[k^α h^(β - 1)] == Log[n / sh] // PowerExpand
Solve[hdot0, Log[k]]
D[Log[k] /. First@%, Log[h]]
```

Out[∘]= $(-1 + \beta)\,\text{Log}[h] + \alpha\,\text{Log}[k] == \text{Log}[n] - \text{Log}[sh]$

Out[∘]= $\left\{\left\{\text{Log}[k] \to \frac{\text{Log}[h] - \beta\,\text{Log}[h] + \text{Log}[n] - \text{Log}[sh]}{\alpha}\right\}\right\}$

Out[∘]= $\frac{1 - \beta}{\alpha}$

Characterize the Data

MRW (p.419) proxy the rate of human-capital accumulation by the fraction of the eligible population (aged 12 to 17) enrolled in secondary school multiplied by the fraction of the working-age population that is of school age (aged 15 to 19). This is the SCHOOL column of their data appendix (p.436). Our comparative statics results suggest that increasing the rate of investment in human capital should increase per capita income. To explore this, take a look at the correlation matrix.

In[∘]:=
```
dsI[Correlation@*Normal@*Values, {"IoY", "SCHOOL", "agepop", "Y1985"}]
```

Out[∘]=

1.0	0.589224	−0.393421	0.665647
0.589224	1.0	−0.380738	0.700287
−0.393421	−0.380738	1.0	−0.606623
0.665647	0.700287	−0.606623	1.0

Out[∘]//TableForm=

	sk	sh	n	Y/N
sk	1.	0.589224	-0.393421	0.665647
sh	0.589224	1.	-0.380738	0.700287
n	-0.393421	-0.380738	1.	-0.606623
Y/N	0.665647	0.700287	-0.606623	1.

Use a scatter plot to visually examine the correlation of Y/N with the new s_h variable. (We have already considered the other correlations.)

In[∘]:=
```
dsI[ListPlot, {"SCHOOL", "Y1985"}]
```

Out[∘]=



Least Squares

Once again, recode the data to prepare for the regression analysis. This is almost the same as before.

In[∘]:=
```
recoderMRW = <|
    "ln`sk" → Log[#IoY / 100],
    "ln`sh" → Log[#SCHOOL / 100],
    "ln`n" → Log[0.05 + #agepop / 100],
    (* recall MRW notation difference (here n=d+gA+gN) *)
    "ln`y" → Log[#Y1985]
    |> &;
dsI`recodeMRW = dsI[All, recoderMRW]
```

Out[∘]=

ln`sk	ln`sh	ln`n	ln`y
−1.42296	−3.10109	−2.57702	8.38275
−1.26231	−3.54046	−2.50104	8.20822
−2.05573	−3.38139	−2.64508	7.69166
−2.91877	−4.50986	−2.6173	6.41017
−2.08747	−3.77226	−2.37516	7.44073

rows 1–5 of 75

```
In[ ]:= dsI`fitMRW =
         LinearModelFit[Normal@Values@dsI`recodeMRW, {ln`sk, ln`sh, ln`n}, {ln`sk, ln`sh, ln`n}]
        dsI`fitMRW["ParameterTable"]
        dsI`fitMRW["AdjustedRSquared"]
```

Out[ ]= FittedModel[≪4≫ + 0.700367 ≪5≫]

Out[ ]=

	Estimate	Standard Error	t-Statistic	P-Value
1	7.79131	1.19242	6.53401	8.30147×10^{-9}
ln`sk	0.700367	0.150583	4.65105	0.0000148757
ln`sh	0.730549	0.0952292	7.67148	6.79223×10^{-11}
ln`n	−1.49978	0.403216	−3.71955	0.00039564

Out[ ]= 0.771425

A few results immediately stand out.

- s_h significant
- increases $\overline{R}^2$ to .78
- lowers estimated α to a more reasonable value (closer to 1/3)

```
In[ ]:= Solve[(β / (1 - α - β)) == 0.730549
           && (α / (1 - α - β)) == 0.700367, {α, β}] // Quiet
```

Out[ ]= {{α → 0.288108, β → 0.300524}}

Furthermore, we can show that this extended model does not reject restriction that coefficients sum to zero.

```
In[ ]:= dsI`rfitMRW = LinearModelFit[Normal@Values@dsI`recodeMRW,
           {ln`sk - ln`n, ln`sh - ln`n}, {ln`sk, ln`sh, ln`n}];
        dsI`rfitMRW["ParameterTable"]
        StringTemplate["adj R2: ``"]@dsI`rfitMRW["AdjustedRSquared"]
```

Out[ ]=

	Estimate	Standard Error	t-Statistic	P-Value
1	7.96624	0.154438	51.5821	1.33038×10^{-58}
−ln`n + ln`sk	0.709078	0.137653	5.1512	2.17107×10^{-6}
−ln`n + ln`sh	0.733038	0.0930925	7.87429	2.63155×10^{-11}

Out[ ]= adj R2: 0.774531

In conclusion, adding human capital improves the empirical performance of the Solow model and thereby calls into question the need for the endogenous growth models.

A possible objection is that this analysis uses a somewhat dubious proxy for aggregate human capital. The focus is almost exclusively on schooling rather than training. This is mainly due to data limitations, but less excusable is the attention frequently paid to school enrollment rates. These rates were initially regarded as one of the more robust and satisfactory variables in the growth literature, but it is worth remembering that they are likely to be positively correlated with initial efficiency, so the results could often be spurious. Certainly it has been much harder to find an effect of human capital in panel data studies, although it is also true that too few researchers carefully discuss the specification. (Rather optimistically, they tend to expect a change in school enrollments to raise growth almost instantly.) Equally importantly, there are some conceptual difficulties with the use of school enrollment data. Only rarely do these rates correspond well to the human capital variables highlighted in theoretical models. In many empirical growth papers it is not clear whether school enrollment rates are intended to represent a flow of investment in human capital, or its stock. In practice these rates may be a poor proxy for either, and given that data on average years of schooling in the population or labour force is now available, the continuing use of school enrollment figures has little to recommend it.

Understanding the Change in Results

Recall the steady-state value of relative human capital.

$$h_{ss} = \left(s_k^{\alpha}\, s_h^{1-\alpha} / n\right)^{1/(1-\alpha-\beta)} \tag{14}$$

A logarithmic transformation produces

$$(1-\alpha-\beta)\,\text{Log}[h_{ss}] == \alpha\,\text{Log}[s_k] + (1-\alpha)\,\text{Log}[s_h] - \text{Log}[n] \tag{15}$$

so that

$$\text{Log}[s_h] == -((1-\alpha-\beta)/(1-\alpha))\,\text{Log}[h_{ss}] - \frac{\alpha}{1-\alpha}\,\text{Log}[s_k] + \frac{1}{1-\alpha}\,\text{Log}[n] \tag{16}$$

is a steady state relationship. Recalling the regression equation

$$\ln[Y/N] = a_t + \frac{\alpha}{1-\alpha-\beta}\,\text{Log}[s_k] + \frac{\beta}{1-\alpha-\beta}\,\text{Log}[s_h] - \frac{\alpha+\beta}{1-\alpha-\beta}\,\text{Log}[n] \tag{17}$$

Substituting for s_h yields

$$\text{Log}[Y/N] = a_t + \frac{\alpha}{1-\alpha}\,\text{Log}[s_k] - \frac{\alpha}{1-\alpha}\,\text{Log}[n] + \frac{\beta}{1-\alpha}\,\text{Log}[h_{ss}] \tag{18}$$

Except for the last term, this is identical to the relation derived from the traditional production function (with no human capital). From this perspective, the original estimation had an omitted regressor.

Now the steady state solution for h implies that it is positively correlated with s_k and negatively correlated with n. Omitting h would therefore tend to increase the magnitude of the coefficients on the other two regressors. This bias can explain our earlier puzzle about the coefficient magnitudes when we estimate the traditional neoclassical growth model.

A focus on potential bias can be problematic for the new results as well, however. For example, as noted by Grossman and Helpman (1994), if low Y/N causes high n this can explain their negative coefficient on n. (Similarly if investment is encouraged by growth.)

MRW draw attention to one particularly dramatic finding. They claim that around 80% of the international variation in per capita incomes can be explained using just three variables: population growth, and investment rates for physical and human capital. The corollary is that differences in technical efficiency can have only a small role to play in explaining cross-country income variation. However, their approach makes two controversial assumptions, that investment rates are exogenous to the level of income and uncorrelated with efficiency. More recently, Peter J. Klenow and Andrés Rodríguez-Clare

(1997) have drawn attention to another underlying problem with the MRW finding. The human capital variable that MRW use only captures variation in secondary schooling. Since it ignores primary schooling, it tends to exaggerate the variation in human capital across countries. When Klenow and Rodríguez-Clare correct for this, they find that MRW's model only explains around half the variation in incomes, leaving a central role for technology differences.

Some Problems

Look just at the OECD countries.

```
In[∘]:= dsO`recodeMRW = <|
        "sk" → #IoY / 100,
        "sh" → #SCHOOL / 100,
        "d+n" → 0.05 + #agepop / 100, (* recall notation difference (n=gA+gN) *)
        "lny" → Log[#Y1985]
        |> & /@ dsO;
      dsO`fitMRW =
        LinearModelFit[Normal@Values@dsO`recodeMRW, {Log[sk], Log[sh], Log[dn]}, {sk, sh, dn}];
      dsO`fitMRW["ParameterTable"]
```

Out[∘]=

	Estimate	Standard Error	t-Statistic	P-Value
1	8.63689	2.21427	3.90057	0.0010481
Log[sk]	0.276134	0.388924	0.709994	0.486805
Log[sh]	0.767571	0.293298	2.61704	0.017462
Log[dn]	−1.07551	0.756005	−1.42262	0.171947

```
In[∘]:= dsO`fitMRW["AdjustedRSquared"]
Out[∘]= 0.244411
```

Problem: If we restrict our attention to the OECD countries, our $\overline{R}^2 = .24$ and the coefficients are not significant!

They report similar numbers: if the sample is restricted to only the 22 OECD countries, the adjusted R^2 falls to .28. The differences in n and I/Y between rich and poor countries drives their results.

Problems: There are still some problematic stylized facts. As Lucas (1988) notes, international wage differentials and patterns of migration don't seem to fit the neoclassical framework. As Romer notes about the Philippines, a worker should not be able to earn a higher wage by migrating to the US. Further, empirically human capital migrates from places where it is scare (and, in a neoclassical setting, should therefore be highly rewarded) to places where it is abundant.

Problem: If one follows them in assuming that TFP has grown at a rate of 2% a year in all countries since 1960, what does one make of the many countries which have grown at less than 2% a year over this period? The model implies that unless these countries have received a large negative shock to technology they are converging to their steady state from above. (I.e., such a country must have exceeded its steady state capital stock in the early 1960s and have been running it down subsequently.) This seems unlikely. There are other problems. Economic miracles, such as Japan's post-war growth, are hard to explain by capital accumulation alone. In most frameworks, such a process would have to be accompanied by a steeply falling real interest rate, something which has not been observed. Temple (1999) argues that such problems suggest that the neoclassical revival in growth economics went too far. Understanding the reasons for differences in technical efficiency and TFP growth is essential to the empirical growth project.